

ZY1420A 分段录音播放应用设计

以及语音报时钟的设计原理

ZY1420是广州致远电子有限公司开发的优质微型语音录放模块，ZY1420内部使用ISD1420作为主控芯片，且具备ISD1420的全部优良性能，如大容量的EEROM存储器，消噪的话筒放大器，自动增益调节AGC电路，专用语音滤波电路，高稳定性的时钟震荡电路和语音处理电路。除此以外，ZY1420还对ISD1420的标准外围电路作了优化并全部集成于模块内部。同用户使用标准ISD1420的DIP40封装IC相比较，ZY1420可以提供更加稳定可靠的性能，更低的价格，更方便的使用，同时还可以减小实际的体积。

ZY1420提供6.4K取样频率，取样的语音直接存储到片内的不挥发存储器内部，不需要数字化和压缩的其它手段。直接模拟存储能提供真实自然的语音，音乐，声音，不象其它的固态数字录音质量要受到影响。ZY1420分为A、B两种，它们最大的不同之处在于外观封装不同，可分段数不同，详细看表1。


表1 ZY1420A和ZY1420B主要异同

型号	ZY1420A	ZY1420B
工作电压	4.5V~5.5V	
封装	标准DIP28	单排14脚
总录放时间	20秒	
地址线数目	7根	2根分段控制线
可分段数	160段	4段
每段时间	0.125秒	5秒
录音控制	电平控制	
放音控制	电平控制、脉冲触发	
是否支持操作模式	支持	不支持

同时，致远电子有限公司提供了一整套的ZY1420模块的测试/演示板，对ZY1420A、B款进行测试和功能演示。其电路原理图如下页所示（图1）。

测试/演示板包括了电源、LCD 液晶显示模块、飞利浦的 P87LPC764 低功耗单片机、I²C 的 E²PROM、功放及 ZY1420 的测试接口等。可以同时测试 ZY1420A、B 款。注意为了同时支持 AB 型测试时测试座与测试模块采用最底对齐的方法。其中 LCD 液晶显示模块的控制芯片为 HT1621，其详细资料可以在广州致远电子有限公司网站：<http://www.21cm.com.cn/> 里下载。

ZY1420 使用上来也十分简单，按住录音按键（REC 保持低电平），电路进入录音状态（录音指示 LED 亮），当 REC 变高或录音存储器录满时，电路退出录音状态进入准备状态。轻按 PLAYE 按键，这样给 PLAYE 脚一个低电平脉冲，电路进入触发放音状态，直到放音结束。按下 PLAYL 按键（PLAYL 脚保持为低电平），电路进入电平放音状态，直到 PLAYL 变高或放音结束，电路重新进入准备状态。注意 REC 的优先级大于 PALYE 和 PLAYL。



一、P87LPC764 单片机控制 ZY1420A 语音模块子程序

ZY1420A 主要有三个功能：录音、电平控制放音、脉冲触发放音。那我们可以分别编写三个子程序与之对应。我们参照 ZY1420 的测试 / 演示板来定义 P87LPC764 的 I/O 口。

1. 录音子程序

控制录音主要有两个数据，1、录音的开始地址；2、录音的时间。把这两个数据都控制了，就可以控制录音时把声音录到那儿里头。注意：REC 信号将被延迟 50ms 防止开关抖动引起重复触发。ZY1420 的更详细资料请参阅广州致远电子有限公司网站：<http://www.21cm.com.cn/>。录音子程序原代码如下：

```
*****
;
;名称:RECORD
;功能:录音
;入口:R7(开始地址),R6(H),R5(L)(录音时间,单位:ms)
;    如果 R6,R5>剩余录音时间,则录完为止
*****
RECORD:
    MOV     A,R7
    LCALL   ADDR_OUT           ;设定开始录音地址
    CLR     REC                 ;控制开始录音
    MOV     R7,#50             ;控制延时 50ms
RECORD1:
    LCALL   DELAY1MS
    DJNZ    R7,RECORD1
RECORD2:
    LCALL   DELAY1MS           ;延时 1ms
    JB      RECD,RECORD3       ;录音模块录满结束判断
    DJNZ    R5,RECORD2         ;录音时间控制
    DJNZ    R6,RECORD2
RECORD3:
    SETB    REC                ;释放录音按键
    RET
```

在调用录音子程序时只要给出录音的开始地址和需要录音的时间就可以了。

例如：从 04H 地址开始录 450ms,则为

```
MOV     R7,#04H               ;设置开始地址: 04h
MOV     R6,#HIGH(450)         ;设置录音时间: 450ms
MOV     R5,#LOW(450)
LCALL   RECORD
```

这样就可以调用录音子程序来录音了，而且指明了声音存放的地址。

2. 电平控制放音子程序

电平控制放音中开始地址和播放时间也是必不可少的，只要控制了这两个参数就可以确定播出内容。不过需要注意的是，播放的时候总是从一段的开头开始播放的，当需要分段播放控制时，录音时每一个内容的开头必须从一个段的开头开始。电平控制放音子程序原代码如下：

```
;*****
```

```
;名称:PLAYL
```

```
;功能:电平控制放音
```

```
;入口:R7(开始地址),R6(H),R5(L)(放音时间,单位:ms)
```

```
; 如果 R6,R5>剩余放音时间,则放完为止
```

```
;*****
```

```
PLAYL:
```

```
MOV A,R7
```

```
LCALL ADDR_OUT ;设定开始播放地址
```

```
CLR PLL ;控制播放
```

```
PLAYL1:
```

```
LCALL DELAY1MS ;延时 1ms
```

```
JNB RECLEL,PLAYL2 ;检测语音模块是否播放完毕
```

```
DJNZ R5,PLAYL1 ;播放时间控制
```

```
DJNZ R6,PLAYL1
```

```
PLAYL2:
```

```
SETB PLL ;释放播放按键
```

```
RET
```

在调用电平控制放音子程序时只要给出放音的开始地址和放音的时间就可以了。

例如：从 08H 地址开始放音 5000ms,则为

```
MOV R7,#08H ;设置开始地址：08h
```

```
MOV R6,#HIGH(5000) ;设置放音时间：5000ms
```

```
MOV R5,#LOW(5000)
```

```
LCALL PLAYL
```

这样就可以调用电平控制放音子程序来放出你所需要的内容。

3. 脉冲触发放音子程序

脉冲触发放音与电平控制放音有些不同，脉冲触发放音不能够由单片机来控制放音的时间，只能够控制放音的开始地址。脉冲触发放音开始后就一直播放到遇到结束符或语音芯片的尽头。所以一般在分段语音控制里头不常用。脉冲触发放音子程序原代码如下：

```
;*****
```

```
;名称:PLAYE
```

```
;功能:脉冲触发放音
```

```
;入口:R7(开始地址)
```

```
;*****
```

```
PLAYE:
```

```
MOV A,R7
```

```
LCALL ADDR_OUT ;设定开始播放地址
```

```
CLR PLE ;控制脉冲触发播放
```

```
MOV R7,#50 ;延时 50ms
```

```
PLAYE1:
```

```
LCALL DELAY1MS
```

```
DJNZ R7,PLAYE1
```

```
SETB PLE ;结束控制信号
```

RET

在调用脉冲触发发音子程序时只要给出发音的开始地址就可以了。

例如：从 10H 地址开始发音,则为

```
MOV    R7,#10H                ;设置开始地址：10h
```

```
LCALL  PLAYE
```

这样就可以调用脉冲触发发音子程序来放出你所需要的内容，不过发音会直到语音模块结束，除非遇到结束符。

二、自动报时计时钟

下面用 ZY1420 的 DEMO 板以一个计时钟的例子说明如何进行分段录音和播放。

计时钟的功能为每一分钟报一次时间，由 0 时 0 分 0 秒开始。

要实现分段播放，先要进行录音，录音可以采用高级的声音处理软件，一次把声音灌进语音模块里头，也可以采用分段录音的方法进行分段录制。下面先介绍用我们的功能测试/演示板来分段录音。

计时钟的主要发音的字有：0，1，2，3，4，5，6，7，8，9，十，时，分，秒。设每 0.5 秒读一个字。我们先建立下面的一个表格：

地址	内容	地址	内容
00H	发音"0"	1CH	发音"7"
04H	发音"1"	20H	发音"8"
08H	发音"2"	24H	发音"9"
0CH	发音"3"	28H	发音"十"
10H	发音"4"	2CH	发音"分钟"
14H	发音"5"		
18H	发音"6"		

那我们先来把 ZY1420 的地址设置为 00H (A0~A7 按键都按下)，按下录音键，把 0 录进去，不过得要注意 0 的发音时间不能超过 0.5 秒，不然会在下面录制 1 的时候把 0 覆盖了，最好控制在 450ms 左右。录好了 0 后，把地址设置到 04H，再录 1。就这样把 0 到分钟的发音都录进去了，可以用电平播放的方式先播放一次，看看录好了没有，如果录得不好可以再录一次。这样就可以把声音全部分段录进去了。

录好以后，我们就可以来编写 P87LPC764 的分段播放控制程序了。不过用单片机进行控制的时候地址按键和控制按键都要松开，控制信号由单片机进行处理。

其控制程序的源代码如下：

```
;P87LPC764, 6MHz 晶振, 6CLOCK, 看门狗关闭, 电压检测 3.8V, 复位高电平。
```

```
;I/O 口分配
```

```
PLL      BIT      P0.2      ;电平发音控制
PLE      BIT      P0.1      ;触发放音控制
REC      BIT      P0.0      ;录音控制
RECLEDD  BIT      P1.5      ;ZY1420 录音输出信号
LCD_CS   BIT      P1.1      ;LCD 片选控制
LCD_WR   BIT      P1.0      ;LCD 写控制
LCD_DAT  BIT      P1.3      ;LCD 数据线
```

```
A0      BIT      P1.4      ;ZY1420 地址线
```

```
A1      BIT      P1.6
```

```
A2      BIT      P1.7
```

A3	BIT	P0.3	
A4	BIT	P0.4	
A5	BIT	P0.5	
A6	BIT	P0.6	
A7	BIT	P0.7	
DISP_DATH	DATA	52H	;LCD 串行输出数据高位
DISP_DATL	DATA	53H	;LCD 串行输出数据低位
DISP_CNT	DATA	54H	;LCD 输出数据位数目
MINUTE	DATA	30H	;时间寄存器,分钟
SECOND	DATA	31H	;时间寄存器,秒
ORG	0000H		;复位矢量
JMP	MAIN		
MAIN:			
MOV	SP,#60H		;设定堆栈
MOV	MINUTE,#00H		;时间清 0
MOV	SECOND,#01		
LCALL	DISP_INIT		;LCD 显示初始化
LCALL	DISP_CLR		;清 LCD 显示内容
MOV	R7,#00H		;LCD 显示时间 0
LCALL	DISP_NUM		
LOOP:			
WAIT:			
LCALL	DELAY1S		;延时 1 分钟
DJNZ	SECOND,WAIT		
MOV	SECOND,#01		
MOV	A,MINUTE		;分钟时间加 1
ADD	A,#01H		
DA	A		
MOV	MINUTE,A		
MOV	R7,MINUTE		;LCD 显示时间
LCALL	DISP_NUM		
MOV	A,MINUTE		;提取十位
ANL	A,#0F0H		
JZ	SAY1		
SWAP	A		
LCALL	TALK		;十位发音
MOV	A,#0AH		
LCALL	TALK		; "十"字发音

SAY1:

```
MOV    A,MINUTE
JZ      SAY2
ANL     A,#0FH      ;提取个位
JZ      SAY3
```

SAY2:

```
LCALL   TALK      ;个位发音
```

SAY3:

```
MOV     A,#0BH      ;"分钟"发音
LCALL   TALK
JMP     LOOP
```

;*****

;名称:TALK

;功能: 发音子程序

;入口:A (读的内容的编码)

;*****

TALK:

```
MOV     R7,A
RL       A
MOV     DPTR,#SAYTIME      ;查表发音时间
MOVC    A,@A+DPTR
MOV     R6,A
INC     R6
MOV     A,R7
RL       A
INC     A
MOVC    A,@A+DPTR
MOV     R5,A
INC     R5
MOV     A,R7
MOV     DPTR,#SAYADDR      ;查表发音地址
MOVC    A,@A+DPTR
MOV     R7,A
LCALL   PLAYL      ;控制分段发音
MOV     R7,#30      ;延时 30ms
```

TALK_WAIT:

```
LCALL   DELAY1MS
DJNZ    R7,TALK_WAIT
RET
```

;*****

;名称:DELAY1S

;功能:延时 1 秒子程序

;*****

DELAY1S:

MOV R7,#100

DELAY1S1:

MOV R6,#100

DELAY1S2:

MOV R5,#50

DJNZ R5,\$

DJNZ R6,DELAY1S2

DJNZ R7,DELAY1S1

RET

;*****

;名称:RECORD

;功能:录音

;入口:R7(开始地址),R6(H),R5(L)(录音时间,单位:ms)

; 如果 R6,R5>剩余录音时间,则录完为止

;*****

RECORD:

MOV A,R7

LCALL ADDR_OUT ;设定开始录音地址

CLR REC ;控制开始录音

MOV R7,#50 ;控制延时 50ms

RECORD1:

LCALL DELAY1MS

DJNZ R7,RECORD1

RECORD2:

LCALL DELAY1MS ;延时 1ms

JB RECLE,RECORD3 ;录音模块录满结束判断

DJNZ R5,RECORD2 ;录音时间控制

DJNZ R6,RECORD2

RECORD3:

SETB REC ;释放录音按键

RET

;*****

;名称:PLAYL

;功能:电平控制放音

;入口:R7(开始地址),R6(H),R5(L)(放音时间,单位:ms)

; 如果 R6,R5>剩余放音时间,则放完为止

;*****

PLAYL:

```
MOV    A,R7
LCALL  ADDR_OUT      ;设定开始播放地址
CLR    PLL           ;控制播放
```

PLAYL1:

```
LCALL  DELAY1MS      ;延时 1ms
JNB    RECLE,PLAYL2  ;检测语音模块是否播放完毕
DJNZ   R5,PLAYL1     ;播放时间控制
DJNZ   R6,PLAYL1
```

PLAYL2:

```
SETB   PLL           ;释放播放按键
RET
```

;名称:PLAYE

;功能:脉冲触发放音

;入口:R7(开始地址)

PLAYE:

```
MOV    A,R7
LCALL  ADDR_OUT      ;设定开始播放地址
CLR    PLE           ;控制脉冲触发播放
MOV    R7,#50        ;延时 50ms
```

PLAYE1:

```
LCALL  DELAY1MS
DJNZ   R7,PLAYE1
SETB   PLE           ;结束控制信号
RET
```

;名称:DELAY1MS

;功能:延时

;入口:R6,R5

DELAY1MS:

```
MOV    R4, #248
```

DELAY1MS1:

```
NOP
NOP
DJNZ   R4,DELAY1MS1
RET
```

;*****

;名称:ADDR_OUT

;功能:发送地址

;入口:A(地址)

;*****

ADDR_OUT:

```
RLC    A
MOV    A7,C
RLC    A
MOV    A6,C
RLC    A
MOV    A5,C
RLC    A
MOV    A4,C
RLC    A
MOV    A3,C
RLC    A
MOV    A2,C
RLC    A
MOV    A1,C
RLC    A
MOV    A0,C
RET
```

;*****

; 名称: DISP_INIT

; 功能: LCD 模块初始化

; 入口: 无

; 出口: 无

; 说明:4 背极,1/3 偏置,片内 RC 振荡

;*****

DISP_INIT:

```
MOV    DISP_DATH,#0AH    ;BIAS  4 背极,1/3 偏置
MOV    DISP_DATL,#40H
MOV    DISP_CNT,#11
LCALL  SEND
MOV    DISP_DATH,#06H    ;RC256  片内 256 振荡
MOV    DISP_DATL,#00H
MOV    DISP_CNT,#11
LCALL  SEND
MOV    DISP_DATH,#00H    ;SYS DIS  关振荡器和偏置发生器
MOV    DISP_DATL,#00H
MOV    DISP_CNT,#11
LCALL  SEND
```

```
MOV    DISP_DATH,#00H      ;SYS EN   开振荡器
MOV    DISP_DATL,#40H
MOV    DISP_CNT,#11
LCALL  SEND
MOV    DISP_DATH,#00H      ;LCD ON   开显示
MOV    DISP_DATL,#10C0H
MOV    DISP_CNT,#11
LCALL  SEND
RET
```

```
;*****
```

```
; 名称: DISP_CLR
```

```
; 功能: 清空显示区
```

```
; 入口: 无
```

```
; 出口: 无
```

```
;*****
```

```
DISP_CLR:
```

```
MOV    R7,#32
```

```
DISP_CLR1:
```

```
MOV    A,R7
```

```
ADD    A,#3FH
```

```
MOV    DISP_DATH,A
```

```
MOV    DISP_DATL,#00H
```

```
MOV    DISP_CNT,#12
```

```
LCALL  SEND
```

```
DJNZ   R7,DISP_CLR1
```

```
RET
```

```
;*****
```

```
; 名称: DISP_NUM
```

```
; 功能: 把 R7 内容显示到 LCD 上
```

```
; 入口: R7
```

```
; 出口: 无
```

```
;*****
```

```
DISP_NUM:
```

```
MOV    DPTR,#NUMTABLE      ;显示十位
```

```
MOV    A,R7
```

```
ANL    A,#0F0H
```

```
SWAP   A
```

```
JNZ    DISP_NUM1
```

```
MOV    A,#0AH
```

```
DISP_NUM1:
```

```
MOVC   A,@A+DPTR
```

```
MOV    DISP_DATL,A
```

```

MOV    DISP_DATH,#46H
MOV    DISP_CNT,#16
LCALL  SEND

```

```

MOV    A,R7                ;显示个位
ANL    A,#0FH
MOVC   A,@A+DPTR
MOV    DISP_DATL,A
MOV    DISP_DATH,#4CH
MOV    DISP_CNT,#16
LCALL  SEND

```

```

MOV    A,#0AH              ;显示小数位
MOVC   A,@A+DPTR
MOV    DISP_DATL,A
MOV    DISP_DATH,#4EH
MOV    DISP_CNT,#16
LCALL  SEND
RET

```

; 名称: SEND

; 功能: DISP_DAT 的内容送到 LCD 模块

; 入口: DISP_DATH,DISP_DATL:输出数据

; DISP_CNT:输出位数

; 出口: 无

SEND:

```

SETB   LCD_DAT              ;片选 LCD
SETB   LCD_WR
SETB   LCD_CS
CLR     LCD_CS
SETB   LCD_DAT              ;输出第一位"1"
CLR     LCD_WR
SETB   LCD_WR

```

SEND1:

```

MOV    A,DISP_DATL          ;输出数据
RLC    A
MOV    DISP_DATL,A
MOV    A,DISP_DATH
RLC    A
MOV    DISP_DATH,A
JC     SEND2
CLR    LCD_DAT

```

```
JMP      SEND3
```

```
SEND2:
```

```
    SETB   LCD_DAT
```

```
SEND3:
```

```
    CLR     LCD_WR
```

```
    SETB   LCD_WR
```

```
    DJNZ    DISP_CNT,SEND1
```

```
    SETB   LCD_CS
```

```
    SETB   LCD_DAT
```

```
    RET
```

```
;*****
```

```
;发音地址控制数据表
```

```
;*****
```

```
SAYADDR:
```

```
;      0   1   2   3   4   5   6   7   8   9   十  分钟
```

```
DB      00H,04H,08H,0CH,10H,14H,18H,1CH,20H,24H,28H,2CH
```

```
;*****
```

```
;发音时间控制数据表,单位:ms
```

```
;*****
```

```
SAYTIME:
```

```
DW      450      ;0 发音时间 450ms
```

```
DW      450      ;1 发音时间 450ms
```

```
DW      450      ;2 发音时间 450ms
```

```
DW      450      ;3 发音时间 450ms
```

```
DW      450      ;4 发音时间 450ms
```

```
DW      450      ;5 发音时间 450ms
```

```
DW      450      ;6 发音时间 450ms
```

```
DW      450      ;7 发音时间 450ms
```

```
DW      450      ;8 发音时间 450ms
```

```
DW      450      ;9 发音时间 450ms
```

```
DW      450      ;十发音时间 450ms
```

```
DW      1000     ;分钟两个字发音时间 1000ms
```

```
;*****
```

```
;LCD 显示表格
```

```
;*****
```

```
NUMTABLE:
```

```
;      0   1   2   3   4   5   6   7   8   9   无
```

```
DB      0EBH,60H,0C7H,0E5H,6CH,0ADH,0AFH,0E0H,0EFH,0EDH,00H
```

```
END
```

这样就可以用做报时了，同时在 LCD 上显示时间。不过倘若要求报时的时间更加准确的话，则要用定时器来作为时间的计数。通过按地址分时播放就可以实现分段播放了，而分段播放的最大优点是可以只要一些基本的声音就可以合成一段话。这样可以节省语音模块的空间，提高产品的灵活性，降低生产成本。分段播放可以用在一些发音的基本元素不多，但组合发音变化比较多的地方。

广州致远电子有限公司对其生产的 ZY1420A、B 型语音模块提供了完善的用户手册、应用介绍指南和技术支持，提供的录放音子程序模块使得你不需再次编写和调试来控制语音模块，使得你的产品能够更早地推出市场。

订购或咨询请致电：(020) 85529859 (020) 85529940 (FAX)

专业服务网站： <http://www.21cm.com.cn/>